

### *Summary*

#### **Main Points**

##### **Setting SAS System Options for Procedure Output**

- SAS system options are instructions that affect your SAS session. All SAS system options have default settings that SAS uses unless you specify otherwise.
- When you create a report, the appearance of the output is determined, in part, by the settings of SAS system options. You can change the settings of SAS system options to alter the way your reports look. Many of the SAS system options for procedure output affect only listing output—not HTML output.

#### **OPTIONS** *option(s);*

- You use the **OPTIONS** statement to modify SAS system options. The **OPTIONS** statement is a global statement.
- When you change the settings of SAS system options, the new settings remain in effect until you change the settings again or until you end your SAS session.
- The **DATE** or **NODATE** option specifies whether or not the date and time appear at the top of your listing output.
- The **NUMBER** or **NONUMBER** option specifies whether or not the page number appears at the top of listing output.
- The **PAGESIZE=** option specifies the number of lines in a page of listing output. The **PAGESIZE =** option has the alias **PS=**.
- The **LINESIZE=** option specifies the number of characters in a line. The **LINESIZE=** option has the alias **LS=**.
- The **CENTER** or **NOCENTER** option specifies whether the output is centered or left-aligned.
- The **DTRESET** or **NODTRESET** option specifies whether or not SAS updates the date and time on every report you create in the same SAS session.
- The **PAGENO=** option specifies the beginning page number for the next page of output.

## Lesson 12: Enhancing Reports

---

### Specifying Titles and Footnotes

```
TITLE <n> 'text';
```

```
FOOTNOTE <n> 'text';
```

- By default, SAS adds a standard title to reports. You can use the TITLE statement to specify a title other than the standard SAS title.
- You can use the FOOTNOTE statement to specify a footnote for your reports.
- You can specify up to 10 titles and 10 footnotes. Titles and footnotes appear in both listing output and HTML output. In both the TITLE statement and the FOOTNOTE statement, you can specify a number from 1 to 10 that indicates the line on which the title or footnote appears. If you don't specify a number, SAS defaults to line 1.
- TITLE and FOOTNOTE statements are global statements. Titles and footnotes that you assign remain in effect until you change them, cancel them, or end your SAS session.
- To cancel a title or footnote, you issue a TITLE or FOOTNOTE statement with the same number as the existing title or footnote and no text; this also cancels any higher-numbered titles or footnotes.
- To cancel all previously defined titles or footnotes, you can specify a null TITLE or FOOTNOTE statement, which has no number and no text.

### Adding Labels and Formats to Reports

- You can change the appearance of variable names in your reports by specifying temporary labels with the LABEL statement in your PROC steps. When you use a LABEL statement within a PROC step, the label applies only to the PROC step that it's part of.

```
SPLIT='split-character'
```

- Remember that to display labels in PROC PRINT reports, you must use the LABEL option in the PROC PRINT statement. Another way to tell SAS to display labels is to add the SPLIT= option in the PROC PRINT statement. The SPLIT= option specifies a split-character that indicates where to wrap the label in the report.
- You can assign a temporary label to a variable that already has a permanent label associated with it. The temporary label always overrides the permanent label in reports.
- You can control how values appear in reports by specifying temporary SAS formats with the FORMAT statement in your PROC steps.

## Lesson 12: Enhancing Reports

---

- You can assign a temporary format to a variable that already has a permanent format associated with it. In reports, the temporary format always overrides the permanent format.

### Creating and Applying User-Defined Formats

```
PROC FORMAT;  
    VALUE format-name value-or-range 1='formatted-value 1'  
        value-or-range 2='formatted-value 2'  
        ...;  
RUN;
```

- If there are no SAS formats that make the changes you want, you can create and apply your own formats to your reports.
- You use the PROC FORMAT step to create your own formats. You can define only one format in a VALUE statement, but you can include multiple VALUE statements in one PROC FORMAT step.
- The names that you specify for your formats must follow the rules for SAS names. Also, you do not put a period after the format name in the VALUE statement, but you do put a period after the format name in the FORMAT statement.
- You can use the keyword HIGH to specify the highest possible value of a variable. You can use the keyword LOW to specify the lowest possible value of a variable. For character variables, the LOW keyword treats missing values as the lowest possible values. For numeric values, LOW does not include missing values.
- You apply your formats in a reporting PROC step, such as a PROC PRINT step, by using a FORMAT statement. You can specify both user-defined formats and SAS formats in the same FORMAT statement.

### Subsetting and Grouping Observations

- Two ways of manipulating observations are subsetting, or selecting only some of the observations in the data set, and grouping observations by the values of one or more variables.
- You can use the WHERE statement in a PROC step to subset observations in a report. When you use the WHERE statement, your output contains only the observations that meet the conditions specified in the WHERE expression.
- You can use the BY statement in reporting procedures to group observations in reports. In a PROC step, the BY statement produces separate sections of the report for each BY group.

## Lesson 12: Enhancing Reports

---

### Sample Code

#### Setting System Options and Specifying Titles and Footnotes

```
options dtreset pageno=1;

title 'Orion Star Sales Salaries';
footnote3 'Confidential';

proc print data=orion.sales;
    var Employee_ID First_Name Last_Name Salary;
run;

title;
footnote;
```

#### Specifying Temporary Labels and Formats

```
proc print data=orion.sales split='*';
    var Employee_ID Job_Title Salary;
    label Employee_ID='Sales ID'
           Job_Title='Job Title'
           Salary='Annual*Salary';
    format Salary dollar10.0;
run;
```

## Lesson 12: Enhancing Reports

---

### Creating and Applying User-Defined Formats

```
proc format;
  value $ctryfmt 'AU' = 'Australia'
                'US' = 'United States'
                other= 'Miscoded';
  value tiers low-<50000 = 'Tier 1'
              50000-100000 = 'Tier 2'
              100000-high= 'Tier 3';
run;

proc print data=orion.sales label;
  var Employee_ID Salary Country Hire_Date;
  label Employee_ID='Sales ID'
         Hire_Date='Hire Date';
  format Salary dollar10.0
         Hire_Date monyy7.
         Country $ctryfmt.
         Salary tiers.;
run;
```

### Subsetting Observations in a Report

```
proc print data=orion.sales;
  var Last_Name Job_Title Country Salary;
  format Salary dollar10.0;
  where Salary>75000;
  title 'Sales Salaries Over $75,000';
run;

title;
```

### Grouping Observations in a Report

```
proc print data=orion.sales;
  by Country descending Gender;
run;
```